public abstract aspect HWServerDistribution {

/**

 * Declare the model classes serializable

 */

declare parents : healthwatcher.model..* implements java.io.Serializable;

/**

 * Creates main method to be able to start the server

 */

public static void HealthWatcherFacade.main(String[] args) {


try {



HealthWatcherFacade.getInstance();


} catch (Exception e) {



e.printStackTrace();


}

}

/**

 * Publishes the server when it starts

 */

protected pointcut serverStart() : execution(static void HealthWatcherFacade.main(String[]));

protected void facadeExceptionHandling(Throwable exception) {


System.out.println("**************************************************");


System.out.println("* Error during server initialization!            *");


System.out.println("* The exception message is:                      *");


System.out.println("      " + exception.getMessage());


System.out.println("* You may have to restart the server or registry.*");


System.out.println("**************************************************");


exception.printStackTrace();

}
}
public abstract aspect HWClientDistribution {

// Makes soft the communication exceptions in the clients

declare soft : CommunicationException : call(* IteratorDsk+.*(..));
    /**
     * Redirects the facade methods call to the distributed instance methods using reflection
     */

protected pointcut facadeCalls() : call(* IFacade+.*(..)) && ! call(static * *.*(..)) && this(CommandServlet+);
    protected void initExceptionHandling(Throwable exception) {
    
String error =  "<p>****************************************************<br>" +
                 "Error during servlet initialization!<br>" +
                 "The exception message is:<br><dd>" + exception.getMessage() +
                 "<p>You may have to restart the servlet container.<br>" +
                 "*******************************************************";
        System.out.println(error);
    }
}
public aspect RMIClientDistribution extends HWClientDistribution {


// Makes soft remote exceptions raised in this aspect


declare soft : RemoteException : within(HWClientDistribution+);


/**


 * Keeps the remote instance


 */


protected IRemoteFacade facade = null;


/**

     * Redirects the facade methods call to the distributed instance methods using reflection

     */


Object around() : facadeCalls() {


    Object obj = MethodExecutor.invoke(


    
getRemoteFacade(), thisJoinPoint.getSignature().getName(), thisJoinPoint.getArgs());


    if (obj instanceof LocalIterator) {




IteratorRMITargetAdapter iteratorTA = new IteratorRMITargetAdapter((LocalIterator) obj, 3);




return new IteratorRMISourceAdapter(iteratorTA, (LocalIterator) obj, 3);


    }


    return obj;

    }

    /**

     * Retrieves the remote instance, creating if needed

     */


private synchronized Remote getRemoteFacade() {        



if (facade == null) {




try {


            System.out.println("About to lookup...");


            facade = (IRemoteFacade) java.rmi.Naming.lookup("//" + healthwatcher.Constants.SERVER_NAME + "/" + healthwatcher.Constants.SYSTEM_NAME);


            System.out.println("Remote DisqueSaude found");


        } catch (java.rmi.RemoteException rmiEx) {


            initExceptionHandling(rmiEx);


        } catch (java.rmi.NotBoundException rmiEx) {


            initExceptionHandling(rmiEx);


        } catch (java.net.MalformedURLException rmiEx) {


            initExceptionHandling(rmiEx);


        }



}



return facade;

    }

}
public aspect RMIServerDistribution extends HWServerDistribution {


/**


 * Forces the facade to implement a remote interface, needed for RMI


 */


declare parents: HealthWatcherFacade implements IRemoteFacade;


/**


 * Publishes the server when it starts


 */


void around() : serverStart() {



try {




System.out.println("Creating RMI server...");




UnicastRemoteObject.exportObject(HealthWatcherFacade.getInstance());




System.out.println("Object exported");




System.out.println(healthwatcher.Constants.SYSTEM_NAME);




java.rmi.Naming.rebind("//" + healthwatcher.Constants.SERVER_NAME + "/"






+ healthwatcher.Constants.SYSTEM_NAME, HealthWatcherFacade.getInstance());




System.out.println("Server created and ready.");



} catch (java.rmi.RemoteException rmiEx) {




facadeExceptionHandling(rmiEx);



} catch (java.net.MalformedURLException rmiEx) {




facadeExceptionHandling(rmiEx);



}


}

}
